
Introduction


It’s not unusual to design applications that have lists of descriptive strings associated with them. A compiler, for example (yes, I know it’s a rather specialized application), might need to generate hundreds of different error messages. Additionally, it’s not unusual for the same message to be generated in a number of different contexts. For the purposes of this column, let’s call such a list an associative list. 





One implementation technique is to construct a pool dictionary and associate the strings with a host of different error message variables. Another is to construct a dictionary that maps symbols to strings and associate the dictionary with a class variable. It is also possible to construct a method that returns the dictionary by using a compile-time constructor as shown below: 





	allMessages	


		^##(Dictionary new


			at: #missingComma 


				put: 'missing comma';


			at: #endOfFile


				put: 'unexpected end';


			yourself)





Personally, I prefer this latter approach because all the information is visible in the method. But more important, I don’t need to do anything special when I file out the code. 





A more typical application might provide status information in a special one-line read-only text pane at the bottom. The status line might be used to indicate that the application is idle (waiting for the user to do something), reading from a file, writing to a file, logging on to a database, … In this case, we have a use for another simple associative list.





As I am prototyping such an application, I don’t typically know what the list should be in advance. When I add a method to provide some new functionality, I might suddenly find the need to invent a new status string. So I’m forced to leave the method I’m in the process of writing (or finish it first) in order to extend the list. Then I’ll come back and continue the prototyping direction that I was pursuing earlier.





Having recently written on the topic of need driven designs [1], I found myself wishing I didn’t have to leave my main train of thought in order to extend the list. Why can’t the system help me make the extension somehow! To tell you the truth, I don’t really care whether the list is extended while I’m developing the code or when I test it as long as it doesn’t disturb my train of thought. When I implement a method that changes the status, it will generally contain a message such as the following:





	self status: #idling





When the application is running, it periodically updates itself. To retrieve the string that needs to be posted in the status pane, the application might end up executing something like


	


	self statusStringFor: self status





Clearly, method statusStringFor: can’t possible work if we haven’t extended the list. So why not make this method intelligent. Why not make it responsible for extending the list. Actually, that’s not all that difficult to do if the list is kept in dictionary referenced by a class variable. But as I argued before, this isn’t as nice a solution as a method that contains all the information. I have something more interesting in mind. By an intelligent method, I mean one that is sophisticated enough to modify and recompile itself.


Preparatory Work


The hardest part about writing an intelligent method is finding out how to compile the method in a non-browser context. The technique that I used to find out how to do this in IBM Smalltalk was to subclass the browser EtClassesBrowser and compile a dummy method. Since EtClassesBrowser contained a method menuFileSave, I wrote my own version as follows:





	menuFileSave


		self halt.


		super menuFileSave





Then I used the debugger to trace how the browser compiles the method. The sequence of messages tracked is detailed in Listing 1. Three aspects had to be resolved: 


Since we wish to compile in a non-browser context, we must replace the “notifying:” parameter by some other suitable object(Transcript seemed to work.


We had to find a substitute for the browser’s instance variable called newApplication. After a lengthy search to find out how to find the application (really the package) associated with a class, we found that it was referred to as the controller(“methodClass controller” “seemed to work.


Finally, we had to find a substitute for the browser’s instance variable called newCategories. By reading the code, we found that nil implied no changes.





The result of all this was a method with header “recompile: source” (see Listing 2) that could be sent to any instance to recompile one of its own method. The method should probably be placed in Object which we would be tempted to do if we had developed a generic framework for manipulating intelligent methods. Since we are illustrating how we might develop one intelligent method “statusStringFor:”, we didn’t feel justified.


Designing a Intelligent Method


An intelligent method ought to be sophisticated enough to know when to recompile itself, to determine how to extend itself, and to regenerate the new source for the extended method. Of course, this presents an interesting problem. How can a method contain the text for itself? Clearly, it can’t. The text for the methods must be contained in the support methods used to reconstruct the intelligent method. So an intelligent method is a self-replicating entity, a very simple type of benign virus. It’s also a form of fractal since it (in one sense) contains itself.





Our intelligent method (see Listing 2) has the form





	statusStringFor: statusSymbol


		| statusMap |


		statusMap := … dictionary literal …


		^statusMap 


			at: statusSymbol 


			ifAbsent: [


				self extendStatusMap: statusMap 


					with: statusSymbol]





Under normal circumstances, the string associated with the status symbol is retrieved from the status map (a dictionary). When the status symbol is missing,  the intelligent method must upgrade itself. In this example, this is done via special message extendStatusMap:with: (see Listing 2). This method performs a number of tasks:





It prompts the software developer for the new string to be associated with the missing status symbol. This is the need-driven aspect that permits the designer to carry on with other design commitments.


It associated this new string with the given status symbol.


It recreates the intelligent method. For this example, the method is regenerated in three pieces: the header portion (up to dictionary literal), the dictionary literal itself, and the trailer portion (everything after the literal).


The new source for the method is recompiled.


Finally, the original query is re-executed to illustrate that it now works. Of course, we could have just as easily returned newString instead.


Conclusions


We defined the term “intelligent method” to mean a method that is sophisticated enough to modify and recompile itself when circumstances warrant. We illustrated it in the context of an application that was intended to display status information. Our intent was to have the method continually upgrade itself during development but not during deployment. If we suitably tested the application, there would never be a need for the intelligent method to revise itself after the application was released. 

Listing 1 Message trace of an EtClassBrowser compiling a method.





	menuFileSave …


	self saveChanges …


	self acceptText: methodText from: self …


	self methodTemplateString: methodText from: self …


	self saveMethodSource: methodText notifying: self …


	self saveMethodSource: methodText 


		in: methodClass notifying: self …


	method := self compile: methodText 


		in: methodClass notifying: self….





which finally led to





	method := methodClass basicCompile: methodText 


		notifying: self ifFail: [].


	… 


	method := self


		install: method


		withSource: methodText 


		ifNewAddTo: newApplication


		categorizeIn: newCategories.








Listing 2 Intelligent method “statusStringFor:” and its support methods.





class:									ExperimentalApplication


superclass:							Object


instance variables:			“none”





class methods 





examples


example1


	"ExperimentalApplication example1"


	^ExperimentalApplication new 


		statusFor: #closing





instance methods





the intelligent method


statusStringFor: statusSymbol


	| statusMap |


	statusMap := ##(Dictionary new


		at: #'writing' put: 'writing to file';


		at: #'validating' put: 'validating data';


		at: #'reading' put: 'reading from file';


		at: #'idling' put: 'waiting for work';


		yourself).


	^statusMap 


		at: statusSymbol 


		ifAbsent: [


			self extendStatusMap: statusMap with: statusSymbol]





extending the intelligent method


extendStatusMap: statusMap with: statusSymbol


	"Reconstruct and recompile method #statusFor:."


	| newString stream |





	"Obtain a new status symbol description from the developer."


	newString := System prompt: 'String for status symbol ', 


		statusSymbol printString.


	newString isNil ifTrue: [^'']. "user cancelled"


	statusMap at: statusSymbol put: newString.


 


	"Generate the source for the method and compile it."


	stream := WriteStream on: (String new: 200).


	self regeneratedStatusStartOn: stream.


	self regeneratedStatusConstant: statusMap on: stream.


	self regeneratedStatusEndOn: stream.


	self recompile: stream contents.





	"Try again."


	^self statusFor: statusSymbol





reconstructing the source for the intelligent method


regeneratedStatusStartOn: stream


	stream nextPutAll:  'statusStringFor: statusSymbol


	| statusMap |


	statusMap := '





regeneratedStatusConstant: statusMap on: stream	


	stream nextPutAll: '##(Dictionary new'.


	statusMap keysAndValuesDo: [:key :value |


		stream 


			cr; tab; tab;


			nextPutAll: 'at: '. key storeOn: stream.


		stream


			nextPutAll: ' put: '. value storeOn: stream.


		stream


			nextPutAll: ';'].


	stream cr; tab; tab; nextPutAll: 'yourself)'





regeneratedStatusEndOn: stream


	stream nextPutAll: '.


	^statusMap 


		at: statusSymbol 


		ifAbsent: [


			self extendStatusMap: statusMap with: statusSymbol]'





recompiling source for a method


recompile: source


	| method |


	method := self class 


		basicCompile: source 


		notifying: Transcript 


		ifFail: [self error: 'failed to recompile'].


	self class


		install: method


		asPrivate: false


		withSource: source 


		ifNewAddTo: self class controller


		categorizeIn: nil

